////////////////////////////////////////////////////////////////////////////////
//
//  The MIT License
//
//  Copyright (c) 2008 - 2009. Dimarik
//
//  Permission is hereby granted, free of charge, to any person obtaining a copy
//  of this software and associated documentation files (the "Software"), to deal
//  in the Software without restriction, including without limitation the rights
//  to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
//  copies of the Software, and to permit persons to whom the Software is
//  furnished to do so, subject to the following conditions:
//
//  The above copyright notice and this permission notice shall be included in
//  all copies or substantial portions of the Software.
//
//  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
//  IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
//  FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
//  AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
//  LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
//  OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
//  THE SOFTWARE.
//
////////////////////////////////////////////////////////////////////////////////

package ru.goodcore.logging.targets
{

import flash.external.ExternalInterface;

import ru.goodcore.logging.LogEvent;
import ru.goodcore.logging.LogEventLevel;
import ru.goodcore.utils.NetworkingStatus;
import ru.goodcore.utils.getNetworkingStatus;

/**
 * Firebug is the popular debug plugin for Mozilla Firefox browser.
 * This class is a target for it's console.
 * 
 * @author				Dimarik
 * @version				1.0
 * @playerversion		9
 * @langversion			3.0
 */

public class FirebugTarget extends AbstractLoggingTarget {

	include "../../../../includes/core/Version.as";

	//-------------------------------------------------------------------------
	//
	//  Class constants
	//
	//-------------------------------------------------------------------------

	// Firebug supports 4 log levels only
	public static const ERROR:		String = "error";

	public static const WARN:		String = "warn";

	public static const INFO:		String = "info";

	public static const LOG:		String = "log";

	//-------------------------------------------------------------------------
	//
	//  Class methods
	//
	//-------------------------------------------------------------------------

	/**
	 * Translates LogEventLevel level to the FireBug console log level 
	 *
	 * @param value
	 * @return 
	 * 
	 */
	private static function getFBLogLevel(value:uint):String {
		var result:String = LOG;

		switch (value) {
			case LogEventLevel.FATAL:
			case LogEventLevel.ERROR:
				result = ERROR;
				break;

			case LogEventLevel.WARN:
				result = WARN;
				break;

			case LogEventLevel.INFO:
				result = INFO;

		}

		return result;
	}


	//-------------------------------------------------------------------------
	//
	//  Constructor
	//
	//-------------------------------------------------------------------------

	/**
	 *  Constructor.
	 */
	public function FirebugTarget() {
		super();
		super.formatter.includeLevel = true;
	}

	//-------------------------------------------------------------------------
	//
	//  Variables
	//
	//-------------------------------------------------------------------------

	//-------------------------------------------------------------------------
	//
	//  Properties
	//
	//-------------------------------------------------------------------------

	//-------------------------------------------------------------------------
	//
	//  Overridden methods
	//
	//-------------------------------------------------------------------------

	public override function logEvent(event:LogEvent):void {
		var msg:String = super.formatter.formatMessage(event);

		if ( getNetworkingStatus() == NetworkingStatus.ALL ) {
			
			try {
				ExternalInterface.call("console." + getFBLogLevel(event.level), msg);
			} catch (error:Error) {
				// do nothing
			}

		}

	}

	//-------------------------------------------------------------------------
	//
	//  Event handlers
	//
	//-------------------------------------------------------------------------

}
}
